

MAGIC ワークブック (IEmesh 編)

IEmesh を使って電磁石のモデルを作成し、ELF/MAGIC による磁場計算を行います。

1 IEmesh の概要

IEmesh は、MESH スクリプトファイル `mei` を編集し、形状ファイル `meg` を作成するためのツールです。

IEmesh の名前は Integral Element Method 用の mesh 作成 から来ています。

IEmesh は、次の2つのプログラムを利用しています。

"Mesh3" MESH スクリプト `mei` を形状ファイル `meg` に変換

"Wmap2" 形状ファイル `meg` の画像を表示

これらのプログラムの組み合わせにより、スクリプトの編集から形状の確認までの一連の作業をスムーズに行うことができます。

画面上のテキストエディタにより、MESH スクリプト `mei` を編集することが主な作業になります。

IEmesh の画面




テキストエディタ	MESH スクリプトのコマンドを1行ずつ入力していきます。
コマンドリスト	コマンドの一覧です。リストからコマンドを選択してダブルクリックすると、テキストエディタのカーソル位置にコマンドがコピーされます。ヒントの内容と連動しています。
ヒント	スクリプトの各行の意味を表示します。テキストエディタのカーソル位置のコマンドの意味が自動的に表示されます。

形状ファイル meg の作成

形状ファイル meg を作成するには、▶ ボタンをクリックします。

▶ ボタンをクリックすると、次の2つの処理が行われます。

1. スクリプトファイル mei を形状ファイル meg に変換します。
2. 続いてその形状の画像を Wmap2 によって表示します。



形状ファイルとログの確認

形状ファイル meg の内容を確認するには、『2 形状データ』タブをクリックします。節点の座標値や、要素を構成している節点の番号が読み取れます。

ログファイル meo の内容を確認するには、『3 ログ』タブをクリックします。ログファイルにはエラーメッセージや、スクリプトで使用した変数の値などが出力されます。

形状ファイル meg のテキスト



2 IEmesh の練習

IEmesh を使って、いろいろな形状データを作成してみます。

最後に ELF/MAGIC の計算に使用する形状ファイル Work.meg を作成します。

IEmesh へのショートカットをクリックします。

出力テンプレート: 使用しないを選択します。

OK をクリックします。

● 節点座標

節点座標を作成します。

テキストエディタに次のように入力してください。

```
G30 1 0 0 0
G30 2 2 0 0
G30 3 4 0 0
OG MGR1
```

G30 3 4 0 0 は節点番号3番の座標が (4,0,0) である事を意味します。

OG MGR1 は座標データを meg ファイルに出力することを意味します。


 をクリックして描画します。

Wmap2 が起動して3節点が描画されます。

『2 形状データ』をクリックして座標を確認してください。

『1 スクリプト』をクリックして元の画面に戻ります。

データの確認が終わったら

 をクリックしてデータを保存します。

保存するファイル名は ABC1.mei とします。

● リポート

同じ内容をリポート行を使って記述できます。

入力したデータを消去して、次のように入力してください。





```
R/3
G30 1/1 0/2 0 0
R
OG MGR1
```

 をクリックして描画します。

R/3 は3回リポートを意味します。

1/1 は節点番号です。初期値1、増分1を意味します。(1, 2, 3)

0/2 は節点の X 座標です。初期値0、増分2を意味します。(0, 2, 4)



●変数


同じことを変数を使って記述できます。

入力したデータを消去して、次のように入力してください。

変数を使うとその数字を書き換えるだけで形状を変化させることができます。

```
PUT X = 2
PUT Y = 0
R/3
G30 1/1 0/X Y 0
R
OG MGR1
```

▶ をクリックして描画します。

データの確認が終わったら  をクリックしてデータを上書き保存します。


- PUT の右辺には式が書けます。例 PUT Y = 2 * (X + 1)
- PUT の右辺の数字が違った複数の mei ファイルを一度に作成することもできます。
- GET で三角関数などの計算をすることもできます。

●2重リピート


2重リピートを使用します。

データを次のように変更してください。

```
PUT X = 1.2
PUT Y = 1.4
R/3/2
G30 1/1/10 0/X/0 0/0/Y 0
R
OG MGR1
```



▶ をクリックして描画します。

データの確認が終わったら  をクリックしてデータを上書き保存します。

X座標は1回目のリピートで、Y座標は2回目のリピートで変化します。

1回目のリピート回数は3回、2回目のリピート回数は2回です。

リピートを使わない次の入力例と同じ結果になります。

```
G30 1 0 0 0
G30 2 1.2 0 0
G30 3 2.4 0 0
G30 11 0 1.4 0
G30 12 1.2 1.4 0
G30 13 2.4 1.4 0
OG MGR1
```

リピートは、6重リピートまで可能です。

G30 1/1/10 0/DX/0 0/0/DY 0 は、次のように一部のゼロや/を省略することができます。

G30 1/1/10 0/DX 0//DY

●要素

要素を作成します。

次のように 6 行のデータを追加してください。


```

PUT X = 1.2
PUT Y = 1.4
R/3/2
G30 1/1/10 0/X/0 0/0/Y 0
R

NAME MMT
MID 1
R/2
E40 0 1/1 2/1 12/1 11/1
R
OET

OG MGR1

```



▶ をクリックして描画します。

E40 を使って、四辺形の平面要素を作成します。要素を構成する4つの節点を指定します。

リピート機能により次の2行を入力したのと同じ効果が得られ、2個の要素が作られます。

```
E40 0 1 2 12 11
```

```
E40 0 2 3 13 12
```

E40 0 1/1 2/1 12/1 11/1 の4つのパラメータは、スラッシュ以降が同じです。

この場合、次のように表すこともできます。

```
E40 0 1/1 2= 12= 11=
```

NAME MMT により要素名を指定します。MMT は薄い磁性体板を表す要素の要素名です。

MID は材質番号を指定するためのデータです。

OET は要素データを meg ファイルに出力することを意味します。

●リメッシュ

1個の要素を細かく分割します。

OET の直前に BL40 2 2 3 の1行を追加してください。番号2の要素を2×3個に分割します。

```

PUT X = 1.2
PUT Y = 1.4
R/3/2
G30 1/1/10 0/X/0 0/0/Y 0
R


NAME MMT
MID 1
R/2
E40 0 1/1 2/1 12/1 11/1
R

BL40 2 2 3

OET

OG MGR1

```



▶ をクリックして描画します。

●押し出し

平面要素を押し出して立体要素にします。3行のデータ行を追加してください。


```

PUT X = 1.2
PUT Y = 1.4
R/3/2
G30 1/1/10 0/X/0 0/0/Y 0
R
NAME MMT
MID 1
R/2
E40 0 1/1 2/1 12/1 11/1
R
BL40 2 2 3

VECX 0 0 0 0 0 1
NAME MMB
EX1 1 100 2.4 2

OET
OG MGR1

```



▶ をクリックして描画します。

平面要素を立体要素にする方法には、この押し出しと、回転押し出しの2種類があります。

●サブルーチン

作成したスクリプトをサブルーチンとして利用できます。データを次のように変更してください。

NAME や **MID** はサブルーチンの外に出し、**CALL** の前においています。

CALL 文の引数を変更することにより、寸法や分割数を変更することができます。


```

NAME MMB
MID 1
CALL ABC ( 1.2 1.4 2.4 2 3 2 )
OET
OG MGR1
STOP

SUBR ABC ( X Y Z N1 N2 N3 )
R/3/2
G30 1/1/10 0/X/0 0/0/Y 0
R
R/2
E40 0 1/1 2/1 12/1 11/1
R
BL40 2 N1 N2
VECX 0 0 0 0 0 1
EX1 1 100 Z N3
RETURN

```

▶ をクリックして描画します。

データの確認が終わったら  をクリックしてデータを上書き保存します。

●DMEG

DMEGにより作成した形状ファイルを読み込むことができます。


IEmesh を閉じてください。

IEmesh へのショートカットをクリックします。

出力テンプレート:使用しないを選択します。

OKをクリックします。次のデータを入力します。


```
DMEG ABC1.MEG 1
OET
OG MGR1
```

 をクリックしてデータを保存します。

保存するファイル名を ABC2.mei とします。

ABC2.mei の保存場所は、AB1.meg と同じフォルダにしてください。

 をクリックして描画します。



読み込んだデータに、新しいデータを追加します。

データには物質データと、空間データがあります。

物質データは OET 行の前に追加します。

空間データは、データを一旦メモリから消去して、新しく作り直します。(変数は消去されません。)


ここでは、空間データを入力します。

次のように入力してください。


```
DMEG ABC1.MEG 1
OET
OG MGR1

CG & CE

PUT L = 5
R/2/2
G30 1/1/2 -HL/L -HL//L 3.0
R
NAME MCO
E40 2 1 2 4 3
BL40 1 5 5
OET
OG MGR2
```



 をクリックして描画します。

データの確認が終わったら  をクリックしてデータを上書き保存します。

HL は L の半分を意味します。

CG & CE はメモリデータの消去を意味します。(変数は消去されません。)

空間データを作成して、最後に形状ファイル meg に出力します。

以上で形状ファイル ABC2.meg ができました。

IEmesh に登録されているサブルーチンを使って、ELF/MAGIC の計算に使用する Work.meg ファイルを作成します。

●データベースの利用

IEmesh へのショートカットをクリックします。
出力テンプレート: 使用しないを選択します。
OK をクリックします。

プルダウンメニューの編集(E)をクリックします。
サブルーチン(S)をクリックします。

- ① SUBR BLOCK (X Y Z I J K) をダブルクリックします。
- ② CALL BLOCK (X Y Z I J K) をダブルクリックします。


テキストエディタには、次のスクリプトが表示されます。


```
CALL BLOCK ( X Y Z I J K )
*****
SUBR BLOCK ( X Y Z I J K )
R/2/2/2
G30 1/1/2/4 -HX/X -HY//Y -HZ///Z
R
E80 0 1 2 4 3 +4
BL80 1 I J K 1. 1. 1.
RETURN
```

サブルーチンのダイアログボックスをいったん閉じます。

CALL 文の引数を数字や変数で置き換えます。
さらに、要素名や材質番号を指定します。
形状ファイル meg への出力も指定します。
次のように入力してください。

```
NAME MMB
MID 1
CALL BLOCK ( 10 10 20 4 4 8 )
OET
OG MGR1
STOP
*****
SUBR BLOCK ( X Y Z I J K )
R/2/2/2
G30 1/1/2/4 -HX/X -HY//Y -HZ///Z
R
E80 0 1 2 4 3 +4
BL80 1 I J K 1. 1. 1.
RETURN
```

▶ をクリックして描画します。
データの確認が終わったら  をクリックしてデータを保存します。
ファイル名は Work.mei とします。



引き続き、2つのサブルーチンを入力します。

OETの前に1行、改行だけの1行を挿入し、この位置にカーソルを合わせます。
サブルーチンのCALL文はこの位置に挿入されます。

プルダウンメニューの編集(E)をクリックします。

サブルーチン(S)をクリックします。

コイルの形状を作るために、サブルーチン COILO を使用します。

① SUBR COILO (RS RL Z N)をダブルクリックします。

② CALL COILO (RS RL Z N)をダブルクリックします。

磁場強度等高線を描く要素を作るために、サブルーチン PLANE を使用します。

① SUBR PLANE (X Y I J)をダブルクリックします。

② CALL PLANE (X Y I J)をダブルクリックします。

この時点で、テキストエディタには次のデータが出力されています。

```

NAME MMB
MID 1
CALL BLOCK ( 10 10 20 2 2 4 )
CALL COILO ( RS RL Z N )
CALL PLANE ( X Y I J )
OET
OG MGR1
STOP
*****
SUBR BLOCK ( X Y Z I J K )
R/2/2/2
G30 1/1/2/4 -HX/X -HY//Y -HZ///Z
R
E80 0 1 2 4 3 +4
BL80 1 I J K 1. 1. 1.
RETURN
*****
SUBR COILO ( RS RL Z N )
PUT D = RL - RS
R/2/2
G30 1/1/2 RS/D 0. -HZ//Z
R
E40 0 2 1 3 4
VECX 0 0 0 0 0 1
EXO 1 1 360 N
RETURN
*****
SUBR PLANE ( X Y I J )
R/2/2
G30 1/1/2 -HX/X -HY//Y 0
R
E40 0 1 2 4 3
BL40 1 I J 1.0 1.0
RETURN

```

CALL文の引数に寸法や分割数を入力します。

CALL COILO (10 15 10 12)

CALL PLANE (20 20 10 10)

STOP までの部分を、次のように変更してください。

```

NAME MMB
MID 1
CALL BLOCK ( 10 10 20 4 4 8 )
STORE 10000


NAME MCL
MID 2
CALL COILO ( 10 15 10 12 )
STORE 10000

NREN & EREN
OL MGSC 0.001
OET
OG MGR1
CG & CE


NAME MCO
MID 3
CALL PLANE ( 20 20 10 10 )
GMO 1 10000 0 0 0 15

OET
OG MGR2
STOP
*****
SUBR BLOCK ( X Y Z I J K )
(以下省略)

```



▶ をクリックして描画します。

データの確認が終わったら  をクリックしてデータを上書き保存します。

STORE は複数のサブルーチンを使うときには欠かせません。

STORE 10000 の入力により、作成した 1~10000 番までのデータの番号が 10001 番以降の番号にリナンバされます。さらにデータを作るときには、もう一度 1 番からデータが作成されます。

NREN & EREN により、節点と要素の番号が 1 番からにリナンバされます。

OL MGSC 0.001 により座標値の単位が 1mm となります。

GM 1 10000 0 0 0 15 により、コンター要素を構成する節点をベクトル (0, 0, 15) の方向へ平行移動します。

以上で形状ファイル Work.meg の完成です。

3 制御・物性データの作成


これから、コアの B-H 特性、コイルの電流値など、計算に必要なデータを準備します。

Work.meg ファイルのアイコンを右クリックし、「mai 作成」を選択してください。
すると MaiEditor2 が起動し、図のような画面が現れます。



3.1 コアの特性


ツリーの項目 磁性体(1)をクリックします。磁性体の B-H 特性を入力します。



- ① ツリーの項目から磁性体(1)をクリック
- ② 材質リストから Fe を選択


3.2 コイルの電流値

ツリーの項目 コイル(2)をクリックします。電流の大きさを入力します。




3.3 制御データ

ツリーの項目 制御データををクリックします。空間磁場の計算を指定します。



各項目の入力が終わったら、上書き保存のアイコンをクリックしてください。
MEG ファイルと同じフォルダに Work.mai ファイルが出力されます。
MaiEditor2 を終了します。




4 ELF/MAGIC の実行

Work.mai ファイルのアイコン上で右クリックし、計算実行を選択してください。


ELF/MAGIC の実行が始まります。

計算が終了すると、Wmap2 が起動し、Work.mag ファイル
(結果ファイル)の内容をベクトルとコンターで表示します。



Wmap2 を終了します。

Launcher を終了します。



5 計算結果の表示

5.1 コンターの表示

Work.mag ファイルの内容を Wmap2 で表示します。

Work.mag ファイルのアイコンをダブルクリックすると Wmap2 が起動します。

ベクトルの表示を OFF にします。

① プルダウンメニュー
【解析結果】-[ベクトル]を選択

② 表示対象「要素磁束密度」の
「ベクトルを表示」のチェックをはずします。

③ 表示対象「空間磁束密度」の
「ベクトルを表示」のチェックをはずします。

④ 「OK」ボタンをクリックします。

コンターの間隔を 0.002T に変更します。

① プルダウンメニュー
[解析結果]-[等高線] を選択

② 「等高線の設定」
ボタンをクリック

③ 「間隔を指定」を選択
④ 0.002 を入力
⑤ 「OK」ボタンをクリック
⑥ 「OK」ボタンをクリック

Level	8	1.20000e-002
Level	7	1.10000e-002
Level	6	1.00000e-002
Level	5	9.00000e-003
Level	4	8.00000e-003
Level	3	7.00000e-003
Level	2	6.00000e-003
Level	1	5.00000e-003

5.2 ベクトルの表示

① プルダウンメニュー
[解析結果]-[ベクトル] を選択

③ 表示対象「要素磁束密度」
「ベクトルを表示」をチェックします。
④ 相対スケール 0.2 を入力
⑤ 表示対象「空間磁束密度」
「ベクトルを表示」をチェックします。
④ 相対スケール 0.2 を入力
⑤ 「OK」ボタンをクリックします。

ベクトルの長さが 2 倍になります。

以上でデータの作成から結果の確認までの一連の作業を終わります。